set terminal pdfcairo font 'Times New Roman,16'   #Font
set output "1d-jacobi-3p.pdf"     #Output

set key box    #Legend position
set key top right
set key width 0.5
set key on


set xlabel "Problem Size (=2^x)" font "Times New Roman,18"
set ylabel "Gstencils/s" font "Times New Roman,18"
set xrange[-0.3:17.3]
set yrange[0:5.5]

set xtics ( "7" 0, "9" 2, "11" 4, "13" 6, "15" 8, "17" 10, "19" 12, "21" 14, "23" 16)

set grid ytics

set label "L1 Size" at 3.4,0.3 font ",12" tc rgb "gray"
set label "L2 Size" at 6.4,0.3 font ",12" tc rgb "gray"
set label "L3 Size" at 13.4,0.3 font ",12" tc rgb "gray"

set arrow from 4,0 to 4,5.5 nohead lc rgb 'gray' dt 2
set arrow from 7,0 to 7,5.5 nohead lc rgb 'gray' dt 2
set arrow from 14,0 to 14,5.5 nohead lc rgb 'gray' dt 2

plot "../result/1d-jacobi-3p.data.txt"   u 6     w lp  lc "red" pt 9 ps 0.9  axis x1y1 t      columnheader(6),'' \
                               u 3     w lp lc "blue"          pt 5 ps 0.6  axis x1y1 t       columnheader(3),'' \
				u 2  w lp  lc "forest-green"  pt 3 ps 0.9  axis x1y1 t  columnheader(2)
